

# CPCL

## 编程手册

## 目录

|                                   |    |
|-----------------------------------|----|
| 1. 命令列表.....                      | 2  |
| 2. 控制命令.....                      | 3  |
| 3. ! 命令.....                      | 3  |
| 4. PAGE-WIDTH 命令.....             | 5  |
| 5. PRINT 命令.....                  | 6  |
| 6. FORM 命令.....                   | 6  |
| 7. TEXT 命令.....                   | 7  |
| 8. COUNT 命令.....                  | 10 |
| 9. SETMAG 命令.....                 | 11 |
| 10. BARCODE-TEXT 命令.....          | 12 |
| 11. BARCODE 命令.....               | 13 |
| 12. PDF417 (便携式数据文件) .....        | 15 |
| 13. QR Code 命令.....               | 16 |
| 14. BOX 命令.....                   | 19 |
| 15. LINE 命令.....                  | 19 |
| 16. INVERSE-LINE 命令.....          | 21 |
| 17. GAP-SENSE 和 BAR-SENSE 命令..... | 23 |
| 18. LEFT CENTER RIGHT 对齐命令.....   | 24 |
| 19. ABORT 命令.....                 | 24 |
| 20. MULTILINE (ML) 命令.....        | 25 |
| 21. EXPANDED-GRAPHICS 图形命令.....   | 26 |
| 22. CONTRAST 命令.....              | 27 |
| 23. TONE 命令.....                  | 27 |
| 24. SPEED 命令.....                 | 28 |
| 25. SETSP 命令.....                 | 29 |
| 26. UNDERLINE 命令.....             | 30 |
| 27. SETBOLD 命令.....               | 31 |

## 1. 命令列表

| 序号 | 命令                | 说明             |
|----|-------------------|----------------|
| 1  | !                 | 进入控制指令 ✓       |
| 2  | PAGE-WIDTH        | 设置打印页面宽度       |
| 3  | PRINT             | 打印控制内容 ✓       |
| 4  | FORM              | 定位控制指令 ✓       |
| 5  | TEXT              | 添加文字指令 ✓       |
| 6  | COUNT             | 数字自动增减指令 ✓     |
| 7  | SETMAG            | 字符放大指令 ✓       |
| 8  | BARCODE-TEXT      | 条码识别符 ✓        |
| 9  | BARCODE           | 一维条码指令 ✓       |
| 10 | BARCODE PDF-417   | PDF-417 条码指令 ✓ |
| 11 | BARCODE QR CODE   | QR 条码指令        |
| 12 | BOX               | 矩形框指令 ✓        |
| 13 | LINE              | 线段指令 ✓         |
| 14 | INVERSE-LINE      | 反白线段指令 ✓       |
| 15 | BAR-SENSE         | 设置右黑标检测指令 ✓    |
|    | GAP-SENSE         | 设置标签检测指令 ✓     |
| 16 | LEFT              | 设置左对齐指令 ✓      |
|    | CENTER            | 设置中心对齐指令 ✓     |
|    | RIGHT             | 设置右对齐指令        |
| 17 | ABORT             | 清除当前指令 ✓       |
| 18 | MULTILINE (ML)    | 打印多行文本指令 ✓     |
| 19 | EXPANDED-GRAPHICS | 图形命令 ✓         |
| 20 | CONTRAST          | 打印黑度指令 ✓       |
| 21 | TONE              | 替代打印黑度指令 ✓     |
| 22 | SPEED             | 打印速度 ✓         |
| 23 | SETSP             | 更改文本字符之间的间距 ✓  |
| 24 | UNDERLINE         | 下划线指令 ✓        |
| 25 | SETBOLD           | 加粗命令 ✓         |

## 2. 控制命令

标签文件通常以“!”字符作为开头，后接“*x*”偏置参数、“*x*”和“*y*”轴分辨率、标签长度以及要打印的标签数量。包含这些参数的行称为命令起始行。

任何情况下，标签文件都是以命令起始行开头，以“*PRINT*”命令结尾。用于构建具体标签的命令置于这两项命令之间。

空格字符用于分隔命令行中的各个字段。



备注：命令会话中的每一行都必须以回车和换行两种字符结尾。所有打印机命令必须采用大写字符。

## 3. ! 命令

格式：

```
<!> {offset} <200> <200> {height} {qty}
```

其中：

<!>：使用“!”作为控制会话的起始字符。

{offset}：整个标签的横向偏置。此值可以使所有域以指定的单位数量进行横向置。


<200>：横向分辨率（以点/英寸为单位）。

<200>：纵向分辨率（以点/英寸为单位）。

{height}：标签的最大高度。

标签最大高度的计算方法是，先测出从第 1 个黑条（或标签间隙）底部到下一个黑条（或标签间隙）顶部之间的距离。然后从中减去 1/16 英寸（1.5 毫米），所得结果即最大高度。（以点为单位时：对于 203 d.p.i 打印机，减去 12 点；对于 306 d.p.i. 打印机，减去 18 点）

{qty}：要打印的标签数量。最大值 = 1024。




打印机命令示例

输入

```
!0 200 200 210 1
TEXT 4 0 30 40 Hello World
FORM
PRINT
```

输出



#### 4. PAGE-WIDTH 命令

打印机假定页面宽度为打印机的完整宽度。打印会话的最大高度由页面宽度和可用打印内存决定。如果页面宽度小于打印机的完整宽度，则用户可以通过指定页面宽度来增加最大页面高度。

**备注：此命令应在打印会话开始时发出。**

格式：

{command} {width}

其中：

{command}：从下面选择一项：

PAGE-WIDTH（或 PW）：指定页面宽度。

{width}：页面的单位宽度。


PAGE-WIDTH 示例

输入 1：

```
!UTILITIES
SETLP 7 0 15
PW 300
PRINT
```

在打印此文本时，标签内存宽度设置为 300 点。

输出 1:




This text is printed with  
label memory width set to  
300 dots.

输入 2:

```
!UTILITIES
SETLP 7 0 15
PW 200
PRINT
```

在打印此文本时，标签内存宽度设置为 200 点。

输出 2:



This text is printed with label  
memory width set to  
200 dots.

## 5. PRINT 命令

PRINT 命令作为整个命令集的结束命令，将会启动文件打印。在任何情况下（行式打印模式除外），这项命令都必须是最后一条命令。执行 PRINT 命令时，打印机将从控制会话中退出。确保使用回车和换行字符结束此项及所有命令。

格式:

{command}

其中:

{command}: PRINT

## 6. FORM 命令

FORM 命令可以指示打印机在一页打印结束后切换至下一页顶部。

格式:

{command}

其中:

{command}: FORM

在下例中，打印机将在打印标签后执行换页操作。有关设置执行 FORM 命令时打印机操作的信息，请参阅设计收据和列表一节中的 SETFF（设置换页）命令。

示例

输入:

```
!0 200 200 3 1
IN-CENTIMETERS
```

```
CENTER
TEXT 4 1 0 5 Hello World
FORM
PRINT
```

7. TEXT 命令

TEXT 命令用于在标签上添加文本。这项命令及其各衍生命令可以控制使用的具体字体号和大小、标签上文本的位置以及文本的方向。标准常驻字体能够以 90 度的增量旋转，如下例所示。

格式：

```
{command} {font} {size} {x} {y} {data}
```

其中：

{command}：从下面选择一项：

| {command}        | 效果                  |
|------------------|---------------------|
| TEXT (或 T)       | 横向打印文本。             |
| VTEXT (或 VT)     | 逆时针旋转 90 度，纵向打印文本。  |
| TEXT90 (或 T90)   | (同 VTEXT。)          |
| TEXT180 (或 T180) | 逆时针旋转 80 度，反转打印文本。  |
| TEXT270 (或 T270) | 逆时针旋转 270 度，纵向打印文本。 |

- {font}：字体名称/编号。
- {size}：字体的大小标识。
- {x}：横向起始位置。
- {y}：纵向起始位置。
- {data}：要打印的文本。

文本

常驻字体示例

Font#: Size:  
0 0

Aa Bb Cc Dd Ee Ff Gg  
\$01.23456789¢ \$0123

Font#: Size:  
0 3

Aa Bb Cc Dd Ee Ff Gg  
\$01.23456789¢ \$0123

Font#: Size:  
0 6

Aa Bb Cc Dd  
\$01.2345678¢

Font#: Size:  
0 1

Aa Bb Cc Dd Ee Ff Gg  
\$01.23456789¢ \$0123

Font#: Size:  
0 4

Aa Bb Cc Dd  
\$01.2345678¢

Font#: Size:  
1 0

Aa Bb Cc Dd Ee Ff Gg  
\$01.23456789¢ \$0123

Font#: Size:  
0 2

Aa Bb Cc Dd Ee Ff Gg  
\$01.23456789¢ \$0123456789 \$0123456789 \$0123

Font#: Size:  
0 5

Aa Bb Cc Dd Ee Ff Gg  
\$01.23456789¢ \$0123

Font#: Size:  
2 0

A C D  
\$01.23456789¢

Font#: Size:  
0 6

Aa Bb Cc Dd  
\$01.2345678¢

常驻字体示例 (续)



常驻字体示例 (续)



示 例

输入:

!0 200 200 210 1  
TEXT 4 0 200 100 TEXT  
TEXT90 4 0 200 100 T90

```
TEXT180 4 0 200 100 T180
TEXT270 4 0 200 100 T270
FORM
PRINT
```

输出:



## 8. COUNT 命令

COUNT 命令可以用于打印多个标签，其中条码中编码的数字文本域或数字数据将针对每个标签依次递增或者递减。TEXT/BARCODE 命令字符串必须包含此数字数据，将其作为字符串的最后若干字符。数字数据部分最多可以包含 20 个字符，且可以以 ‘-’ 符号作为前缀。增加或减少数字数据时不能以 ‘0’ 为增量或减量。前导零将予以保留。一个标签文件中最多可使用三个 COUNT 命令。一个标签文件中最多可使用三个 COUNT 命令。

递增/递减的数字数据包含在 TEXT 或 BARCODE 命令中，后面紧跟 COUNT 命令

格式:

```
{command} {numeric value}
```

其中:

```
{command}: COUNT
```

{numeric value}: 任何整数值都不能超过 20 个字符。如果希望减小 TEXT/BARCODE 值，则可以在值前添加 ‘-’ 符号。输出结果中将保留前导零。

COUNT 命令示例

输入:

```
!0 200 200 210 3
; Print 3 labels
CENTER
TEXT 4 0 0 50 TESTING 001
COUNT 1
TEXT 7 0 0 100 Barcode Value is 123456789
COUNT -10
BARCODE 128 1 1 50 0 130 123456789
COUNT -10
FORM
PRINT
```

输出:



## 9. SETMAG 命令

SETMAG 命令可将常驻字体放大指定的放大倍数。

格式：

{command} {w} {h}

其中：

{command}：SETMAG

{w}：字体的宽度放大倍数。有效放大倍数为 1 到 16。

{h}：字体的高度放大倍数。有效放大倍数为 1 到 16。



备注：SETMAG 命令在标签打印后仍保持有效。这意味着要打印的下一标签将使用最近设置的 SETMAG 值。要取消 SETMAG 值并使打印机可以使用默认字体大小，请使用 SETMAG 命令，且放大倍数为 0.0。

SETMAG 命令示例

输入：

```
!0 200 200 210 1
CENTER
SETMAG 1 1
TEXT 0 0 0 10 Font 0-0 at SETMAG 1 1
SETMAG 1 2
TEXT 0 0 0 40 Font 0-0 at SETMAG 1 2
SETMAG 2 1
TEXT 0 0 0 80 Font 0-0 at SETMAG 2 1
SETMAG 2 2
TEXT 0 0 0 110 Font 0-0 at SETMAG 2 2
SETMAG 2 4
TEXT 0 0 0 145 Font 0-0 at SETMAG 2 4
; Restore default font sizes
SETMAG 0 0
```

FORM  
PRINT

输出:



## 10. BARCODE-TEXT 命令

*BARCODE-TEXT* 命令用于通过创建条码时所用的相同数据来标记条码。这项命令避免了使用单独文本命令注释条码的必要。文本位于条码下方的中间位置。

使用 *BARCODE-TEXT OFF* (或 *BT OFF*) 可以禁用它。

格式:

{command} {font number} {font size} {offset}

其中:

{command}: *BARCODE-TEXT* (或 *BT*)

{font number}: 注释条码时要使用的字体号。

{font size}: 注释条码时要使用的字体大小。

{offset}: 文本距离条码的单位偏移量。

*BARCODE-TEXT* 示例

输入:

```
!0 200 200 400 1
JOURNAL
CENTER
; Annotate bar codes using font 7 size 0
; and offset 5 dots from the bar code.
BARCODE-TEXT 7 0 5
BARCODE 128 1 1 50 0 20 123456789
VBARCODE 128 1 1 50 40 400 112233445
BARCODE-TEXT OFF
FORM
PRINT
```

输出：



## 11. BARCODE 命令

BARCODE 命令能够以指定的宽度和高度纵向和横向打印条码。  
标准条码

格式：

{command} {type} {width} {ratio} {height} {x} {y} {data}

其中：

{command}：从下面选择一项：

BARCODE（或 B）：横向打印条码。

VBARCODE（或 VB）：纵向打印条码。

{type}：从下表中选择：

| 符号：                            | 用法：                       |
|--------------------------------|---------------------------|
| UPC-A                          | UPCA、UPCA2、UPCA5          |
| UPC-E                          | UPCE、UPCE2、UPCE5          |
| EAN/JAN-13                     | EAN13、EAN132、EAN135       |
| EAN/JAN-8                      | EAN8、EAN82、EAN 85         |
| Code 39                        | 39、39C、F39、F39C           |
| Code 93/Ext.93                 | 93                        |
| Interleaved 2 of 5             | I2OF5                     |
| Interleaved 2 of 5（带 checksum） | I2OF5C                    |
| German Post Code               | I2OF5G                    |
| Code 128（自动）                   | 128                       |
| UCC EAN 128                    | UCCEAN128                 |
| Codabar                        | CODABAR、CODABAR16         |
| MSI/Plessey                    | MSI、MSI10、MSI1010、MSI1110 |
| Postnet                        | POSTNET                   |
| FIM                            | FIM                       |

{width}：窄条的单位宽度。

{ratio}：宽条与窄条的比率。有关相应设置，请参阅附录“E”中的表格。

|            |            |            |
|------------|------------|------------|
| 0 = 1.5 :1 | 20 = 2.0:1 | 26 = 2.6:1 |
| 1 = 2.0 :1 | 21 = 2.1:1 | 27 = 2.7:1 |
| 2 = 2.5 :1 | 22 = 2.2:1 | 28 = 2.8:1 |
| 3 = 3.0 :1 | 23 = 2.3:1 | 29 = 2.9:1 |
| 4 = 3.5 :1 | 24 = 2.4:1 | 30 = 3.0:1 |
|            | 25 = 2.5:1 |            |



备注：附录中的比率为推荐值，旨在帮助您获得最佳效果；当然，您可以指定任意比率。

{height}：条码的单位高度。

{x}：横向起始位置。

{y}：纵向起始位置。

{data}：条码数据。

条码示

例输

入：

```
!0 200 200 210 1
  BARCODE 128 1 1 50 150 10 HORIZ.
  TEXT 7 0 210 60 HORIZ.
  VBARCODE 128 1 1 50 10 200 VERT.
  VTEXT 7 0 60 140 VERT.
  FORM
  PRINT
```

输出：



## 12. PDF417（便携式数据文件）

格式：

```
{command} {type} {x} {y} [XD n] [YD n] [C n] [S n]  
{data}  
<ENDPDF>
```

其中：

{command}：从下面选择一项：

    BARCODE（或 B）：横向打印条码。

    VBARCODE（或 VB）：纵向打印条码。

{type}：PDFra-417

{x}：横向起始位置。

{y}：纵向起始位置。

[XD n]：最窄元素的单位宽度。范围介于 1 至 32 之间，默认值为 2。

[YD n]：最窄元素的单位高度。范围介于 1 至 32 之间，默认值为 6。

[C n]：要使用的列数。数据列不包括起始/终止字符和左/右指示符。范围介于 1 至 30 之间，默认值为 3。

[S n]：安全级别，指示要检测和/或纠正的最大错误量。范围介于 0 至 8 之间，默认值为 1。

{data} 条码数据。

<ENDPDF>：终止 PDF-417。



备注：BARCODE-TEXT 命令不能用于 PDF-417 条码类型。对于任何所需的可人工识读文本，必须使用 TEXT 命令单独输入，如下例所示。

PDF417 示例

输入：

```
!0 200 200 210 1  
B PDF-417 10 20 XD 3 YD 12 C 3 S 2  
PDF Data  
ABCDE12345  
ENDPDF  
T 4 0 10 120 PDF Data  
T 4 0 10 170 ABCDE12345  
FORM  
PRINT
```

输出：



## 13. QR Code 命令

格式:

```
{command} {type} {x} {y} [M n] [U n]
{data}
<ENDQR>
```

其中:

{command}: 从下面选择一项:

BARCODE (或 B): 横向打印条码。

VBARCODE (或 VB): 纵向打印条码。

{type}: QR

{x}: 横向起始位置。

{y}: 纵向起始位置

[M n]: QR Code 规范编号。选项是 1 或 2。QR Code Model 1 是原始规范, 而 QR Code Model 2 则是该符号的经过增强后的形式。Model 2 提供了附加功能, 而且可以自动与 Model 1 进行区分。Model 2 为推荐规范, 是默认值。

[U n]: 模块的单位宽度/单位高度。范

围是 1 至 32。默认值为 6。

{data}: 提供生成 QR Code 所需的信息。请参见下面的示例。

{data} 除了包含实际的输入数据字符串外, 还包含一些模式选择符号。

输入数据类型可以由打印机软件自动识别, 也可以通过手动方式设置。模式选择符号和实际数据之间有一个分隔符 (逗号)。

用于自动选择数据类型的数据字段格式:

<Error Correction Level><Mask No.><Data Input Mode (should be  
“A”)>,<Data Character String>

纠错级别应为以下符号之一:

H - 极高可靠性级别 (H 级);

Q - 高可靠性级别 (Q 级);

M - 标准级别 (M 级);

L - 高密度级别 (L 级)。

掩码号可能会省略, 也可能具有一个值 (介于 0 至 8 之间):

无 - 软件自动选择掩码;

0 至 7 - 使用带有相应编号 (0 至 7) 的掩码;

8 - 无掩码。

用于手动选择数据类型的数据字段格式包含字符模式符号, 采用如下格式:

<Error Correction Level><Mask No.><Data Input Mode (should be “M”)>,  
<Character Mode 1><Data Character String 1>,<Character Mode 2><Data Character  
String 2>,< : > : >,<Character Mode n><Data Character String n>

字符模式符号:

N - 数字;

A - 字母数字;

Bxxxx - 二进制, 二进制模式包含由 2 字节 BCD 代码表示的数据字符的数量

(xxxx)。

K - Kanji

不同的数据字段（带有对应的字符模式符号）由逗号分隔。

如果输入模式设置为“自动”，则无法设置 0x80 至 0x9F 和 0xe0 至 0xFF 的二进制代码。

<ENDQR>: 终止 QR Code。

#### 数据字段格式设置示例

##### 示例 1

Error Correction Level:Standard level <M>  
Mask No.:<None>  
Input mode:Automatic setting <A>  
Data:QR Code  
The {data} field presentation for generating a QR code under the conditions above:  
MA, QR Code

##### 示例 2

Error Correction Level:Ultra high reliability level <H>  
Mask No.:<0>  
Input mode:Manual setting <M>  
Character Mode:Numeric mode <N>  
Data:0123456789012345  
The {data} field presentation:  
HOM, N0123456789012345

##### 示例 3

Error Correction Level:Standard level <M>  
Mask:<None> (Automatic selection)  
Input mode:Manual setting <M>  
Character Mode:Alphanumeric mode <A>  
Data:AC-42  
The {data} field presentation:  
MM, AAC-42

##### 示例 4

Error Correction Level:High density level <L>  
Mask No.:Automatic setting <None>  
Input mode:Manual setting <M>  
Character Mode:Alphanumeric <A>  
Data:QR code  
Character Mode:Numeric <N>  
Data:0123456789012345  
Character Mode:Alphanumeric <A>  
Data:QR CODE  
Character Mode:Binary <B>

Data: qrcode  
The {data} field presentation:  
LM, AQRcode, N0123456789012345, AQRCODE, B0006qrcode

 备注: BARCODE-TEXT 命令不能用于 QR Code。对于任何所需的可人工识读文本, 必须使用 TEXT 命令单独输入, 如下例所示。

QR Code 示例

输入:

```
!0 200 200 500 1
B QR 10 100 M 2 U 10
MA, QR code ABC123
ENDQR
T 4 0 10 400 QR code ABC123
FORM
PRINT
```

输出:



QR code ABC123

 备注: 可人工识读的文本不包含在 QR 代码输出结果中。

## 14. BOX 命令

用户可以使用 BOX 命令生成具有指定线条宽度的矩形。

格式：

```
{command} {x0 } {y0 } {x1} {y1} {width}
```

其中：

{command}：BOX

{x0 }：左上角的 X 坐标。

{y0 }：左上角的 Y 坐标。

{x1}：右下角的 X 坐标。

{y1}：右下角的 Y 坐标。


{width}：形成矩形框的线条的单位宽度。

BOX 命令示例

输入：

```
!0 200 200 210 1  
BOX 0 0 200 200 1  
FORM  
PRINT
```

输出：



备注：输出中显示的文本坐标仅用于说明之目的。

## 15. LINE 命令

使用 LINE 命令可以绘制任何长度、宽度和角度方向的线条。

格式：

```
{command} {x0 } {y0 } {x1} {y1} {width}
```

其中：

{command}： 从下面选择一项：

LINE（或 L）：打印线条。

{x0 }：左上角的 X 坐标。

{y0 }：左上角的 Y 坐标。

{x1}: 以下项的 X 坐标:

- 水平轴的右上角。
- 垂直轴的左下角。

{y1}: 以下项的 Y 坐标:


- 水平轴的右上角。
- 垂直轴的左下角。

{width}: 线条的单位宽度

输入:

```
!0 200 200 210 1
LINE 0 0 200 0 1
LINE 0 0 200 200 2
LINE 0 0 0 200 3
FORM
PRINT
```

输出:



备注: 输出中显示的文本坐标仅用于说明之目的。

## 16. INVERSE-LINE 命令

INVERSE-LINE 命令的语法与 LINE 命令相同。位于 INVERSE-LINE 命令所定义区域内的以前创建的对象黑色区域将重绘为白色，白色区域将重绘为黑色。这些对象可以包括文本、条码和/或图形（包括下载的 .pcx 文件）。INVERSE-LINE 对在其之后创建的对象不起作用，即使这些对象位于该命令的覆盖区域内也是如此。在示例 INVERSE2.LBL 中，在 INVERSE-LINE 命令之后创建的文本字段部分仍然为黑色，因此不可见，即使被放置在 INVERSE-LINE 区域内也是如此。

格式:

{command} {x0 } {y0 } {x1} {y1} {width}

其中:

{command}: 从下面选择一项:

INVERSE-LINE (或 IL): 在现有字段上方打印一个线条以反转图像。

{x0 }: 左上角的 X 坐标。

{y0 }: 左上角的 Y 坐标。

{x1}: 以下项的 X 坐标:  
- 水平轴的右上角。  
- 垂直轴的左下角。  
{y1}: 以下项的 Y 坐标:  
- 水平轴的右上角。  
- 垂直轴的左下角。  
{width}: 反转线的单位宽度。

INVERSE-LINE 命令示例

输入 1:

```
!0 200 200 210 1
CENTER
TEXT 4 0 0 45 SAVE
TEXT 4 0 0 95 MORE
INVERSE-LINE 0 45 145 45 45
INVERSE-LINE 0 95 145 95 45
FORM
PRINT
```


输出 1:



输入 2:

```
!0 200 200 210 1
T 4 2 30 20 $123.45
T 4 2 30 70 $678.90
IL 25 40 350 40 90
T 4 2 30 120 $432.10
FORM
PRINT
```

输出 2:



## 17. GAP-SENSE 和 BAR-SENSE 命令

这两项命令用于指示打印机应采用的页顶检测方式。如果未指定任何命令，则打印机默认使用 BAR-SENSE。未配备间隙传感器的打印机将尝试伪间隙感应。

格式：

{command}

其中：{command}：从下面选择一项：

GAP-SENSE # (0-255)

BAR-SENSE # (0-255)

GAP-SENSE 命令示例：

下例将打印机配置为间隙感应。此外，它还指定页顶到间隙的距离为零。

输入：

```
!UTILITIES
GAP-SENSE
SET-TOF 0
PRINT
```

## 18. LEFT CENTER RIGHT 对齐命令

使用对齐命令可以控制字段的对齐方式。默认情况下，打印机将左对齐所有字段。对齐命令将对所有后续字段保持有效，直至指定了其他对齐命令。

格式：

{command} [end]

其中：

{command}：从下面选择一项：

CENTER：居中对齐所有后续字段。

LEFT：左对齐所有后续字段。

RIGHT：右对齐所有后续字段。

[end]：对齐的结束点。如果未输入参数，则对于横向打印，对齐命令将使用打印头的宽度；而对于纵向打印，对齐命令将使用零（页头）。

对齐示例

输入：

```
!0 200 200 210 1
CENTER 383
TEXT 4 0 0 75 C
LEFT
TEXT 4 0 0 75 L
RIGHT 383
TEXT 4 0 0 75 R
FORM
PRINT
```

输出:



## 19. ABORT 命令

ABORT 命令可以在不打印的情况下终止当前的控制会话。

格式:

{command}

其中:

{command}: ABORT

## 20. MULTILINE (ML) 命令

使用 MULTILINE (ML), 可以以相同字体和行高打印多行文本。

格式:

```
{command} {height}  
  {text} {font} {size} {x} {y}  
  {data}  
  “  
  {data}  
  <ENDMULTILINE>
```

其中:

{command}: MULTILINE (或 ML) - 打印多行文本。

{height}: 每行文本的单位高度。

{text}: 文本命令 (TEXT、VTEXT 等)。

{font}: 字体名称/编号。

{size}: 字体的大小标识。

{x}: 横向起始位置。

{y}: 纵向起始位置。

{data}: 要打印的文本。


<ENDMULTILINE> (或 ENDML): 终止 MULTILINE。

ML 命令示例

输入:

```
!0 200 200 210 1  
ML 47  
TEXT 4 0 10 20  
1st line of text  
2nd line of text  
Nth line of text  
ENDML
```

输出:




可以使用图形命令打印位映射图形。扩展图形数据使用 ASCII 十六进制字符来表示（参见示例）。通过对十六进制数据的等效二进制字符使用 COMPRESSED-GRAPHICS 命令，可以将数据大小减半。如果使用 CG，对于每 8 位图形数据，将会发送一个 8 位字符。如果使用 EG，将使用两个字符（16 位）来传输 8 位图形数据，因此 EG 的效率会减半。但是由于该数据是字符数据，因此比二进制数据更容易处理和传输。

| <code>{command}</code> | <code>{width}</code> | <code>{height}</code> | <code>{x}</code> | <code>{y}</code> | <code>{data}</code> |
|------------------------|----------------------|-----------------------|------------------|------------------|---------------------|
|------------------------|----------------------|-----------------------|------------------|------------------|---------------------|

输入:

PRINT

输出:



## 22. CONTRAST 命令

CONTRAST 命令用于指定整个标签的打印黑度。最亮的打印输出为对比度级别 0。最暗的对比度级别为 3。打印机在开机时的默认对比度级别为 0。必须为每个标签文件指定对比度级别。



备注：为了最大限度地提高打印效率，请始终使用尽可能低的对比度级别。

格式：

{command} {level}

其中：

{command}： CONTRAST

{level}： 对比度级别。

0 = 默认值

1 = 中

2 = 暗

3 = 非常暗

## 23. TONE 命令

TONE 命令可用于替代 CONTRAST 命令来指定所有标签的打印黑度。最亮的打印输出为色调级别 -99。最暗的色调级别为 200。打印机在开机时的默认色调级别为 0。色调级别设置在更改前对所有打印任务保持有效。TONE 和 CONTRAST 命令不能彼此组合使用。

格式：

{command} {level}

其中：

{command}： TONE

{level}： 选择介于 -99 到 200 之间的值。

对比度级别和色调级别的等价等效关系：

对比度 0 = 色调 0

对比度 1 = 色调 100

对比度 2 = 色调 200

对比度 3 = 无等效色调



备注：在使用 Zebra Technologies 生产的无衬纸介质时，建议将 TONE 值设置为 25 以获得最佳打印效果。

## 24. SPEED 命令

此命令用于设置电机的最高速度级别。每一款打印机型号都设置了最低和最高极限速度。SPEED 命令可以在 0 到 5 的范围内选择速度级别，0 表示最低速度。为每一款打印机型号设置的最高速度仅可在理想条件下达到。电池或供电电压、材料厚度、打印黑度、是否使用贴标机、是否使用剥离器以及标签长度等诸多因素均会影响最大极限打印速度



警告：在练习此命令时，用户将会覆盖标签打印的出厂设定速度，这可能会导致打印质量下降。如果使用当前 SPEED 设置影响到打印质量，则应降低打印速度。

格式：

{command} {speed level}

其中：

{command}：SPEED

{speed level}：一个介于 0 到 5 之间的数字，0 表示最低速度。

SPEED 命令示例

输入：

```
!0 200 200 150 1
SPEED 4
TEXT 5 0 0 20 PRINTS AT SPEED 4
FORM
PRINT
```

## 25. SETSP 命令

SETSP 命令用于更改文本字符之间的间距。

格式：

{command} {spacing}

其中：

{command}：SETSP

{spacing}：字符间的单位尺寸。间距的默认值为零。请注意，此命令受单位命令设置的影响。

SETSP 命令示例

输入：

```
!0 200 200 210 1
T 4 0 0 10 Normal Spacing
SETSP 5
T 4 0 0 50 Spread Spacing
SETSP 0
T 4 0 0 90 Normal Spacing
FORM
```

输出:



## 26. UNDERLINE 命令

UNDERLINE 命令用于给文本加下划线。仅当所使用的字体支持下划线时，此命令才会起作用。如果所使用的字体不支持 UNDERLINE，则将忽略此命令。

以下字体支持 UNDERLINE:

GBUNSN24.CPF

GBUNSN24.CPF

格式:

```
{command} {mode}  
{command}: UNDERLINE  
{mode}: 从下面选择一项
```

"on"


"off"

UNDERLINE 示例

输入 (UNDERLINE.LBL):

```
!O 200 200 200 1  
ENCODING GB18030  
UNDERLINE ON  
TEXT GBUNSG24.CPF 0 20 30 Underlined ,t,u  
UNDERLINE OFF  
TEXT GBUNSG24.CPF 0 20 80 Normal ,t,u  
ENCODING ASCII  
UNDERLINE OFF  
PRINT
```

输出



## 27. SETBOLD 命令

*SETBOLD* 命令可使文本加粗并且稍微加宽。*SETBOLD* 命令会采用一个操作数来设置文本变黑的程度。

格式:

```
!U1 SETBOLD {value}
```

其中, {value} 是介于 0 到 5 之间的偏移量。



备注: value} 将采用 通过单位命令设置的单位。

默认单位设置以点为单位。(203 点 = 1 英寸)

如果单位为英寸, 则偏移值的范围为 0-0.0246 英寸。

如果单位为厘米, 则偏移值的范围为 0-0.0625 厘米。

如果单位为毫米, 则偏移值的范围为 0-0.625 毫米。

完成后, 请务必发出 “!U1 SETBOLD 0” 命令以禁用粗体格式。

SET BOLD 命令示例

输入:

```
!U1 SETBOLD 2
```

```
This text is in bold !U1 SETBOLD 0
```

```
but this text is normal.
```

输出:

```
This text is in bold but this text is normal.
```